Introduction to Finite Elements with *FreeFem++*

J. Rafael Rodríguez Galván
Universidad de Cádiz

Máster interuniversitario en Matemáticas. Universidad de Cádiz.

February, 2016

Contents

1 Introduction

1.1 The *FreeFem++* environment

FreeFem++,[1] is a package for *numerical approximation of the solution of PDE* (partial differential equations), both 2D and 3D, by means of the *Finite Element Method* (FEM). *FreeFem++* is composed of:

- An interpreted *programming Language*:
  - Oriented to fast specification of those problems which can be described by means of (linear and steady) PDE, and also to resolution of those problems, using FEM.
  - Allows easy implementation of complex problems (nonlinear, transient,...)
- An *interpreter* for that language.
  - Programs in FreeFem++ are interpreted (not compiled) at runtime. In this sense, FreeFem++ is a *scripting* language (like Python, Matlab/Octave, Perl, and others).
  - *FreeFem++* is *open source/free software* (GNU GPL license).
  - There are different versions: `FreFem++`, `FreeFem++-nw`, `FreeFem++-mpi`,...

FreeFem++ is mainly developed by **F. Hetch**.

[1] `http://www.freefem.org/ff++`

1.2 What can you do with *FreeFem++*?

Some examples related to 2D steady Stokes equations:

$$\begin{cases} -\nu\Delta\mathbf{u} + \nabla p = f \\ \nabla\cdot\mathbf{u} = 0 \\ \qquad\qquad + \text{boundary conditions,} \end{cases}$$

where the unknowns are $\mathbf{u} = (u, v) : \Omega \to \mathbb{R}^2$ (velocity field of fluid) and $p : \Omega \to \mathbb{R}$ pressure in each point of the domain.

- The Stokes/Navier-Stokes equations in Mediterranean sea (video).
- Why I like numerical simulation (as mathematician): it helps you to understand theory (in this video, instability of a numerical scheme).

1.3 Characteristics of the *FreeFem++* Language

- Inspired by *C/C++*.
  - *Similarities*: Syntax, strong typing...
  - *Does not include*: Pointers, object orienting, ...
- *Oriented* to numerical simulation using the finite element method. Possibilities:
  - Definition of the geometry of a problem and 2D/3D meshing. Although *FreeFem++* is not a CAD/CAE environment and then, for complex geometries, it is necessary to use external tools.

- Variety of available *finite elements*: P_k–Lagrange, P_1–bubble, P_1 discontinuous, Raviart-Thomas...
- Flexibility for definition of problems which can be formulated in terms of PDE (and expressed by a *variational formulation*)
- Automation of the task of assembling FEM matrices involved in underlying linear systems, so that this task is transparent to the user.
- Several algorithms for *resolution of those linear systems*: LU, Cholesky, Crout, CG, GMRES, UMFPACK...
- Facilities for *post-processing and 2D/3D visualización*. Although *FreeFem++* no is not specialized in scientific visualización, it can be complemented with external tools for high-quality graphics.
- Many other issues:
  * Excellent documentation (with a plenty of examples and tutorials): `http://www.freefem.org/ff++/ftp/freefem++doc.pdf`.
  * *Matlab*–like matrix manipulation (or Matlab/Octave/Python/Fortran–like).
  * Automatic interpolation between meshes, adaptive refinement,...
  * Parallel (with MPI) version available (`FreeFem++-mpi`) in UNIX systems.

2 Installation and first steps

The *FreeFem++* package includes an interpreter for execution of code (code which is written in *FreeFem++* language) and also some additional tools. But the standard edition does not include an integrated environment (with editor, error feedback, syntax highlighting, etc.). User is allowed to choose his/her preferred editor between different possibilities, as we comment below.

In each operative system, there are different possibilities for the selection of an adequate editor, for instance:

- *Crimson Editor* or *Notepad++* in Windows,
- `Emacs` in GNU/Linux, MacOS or Windows.
- ...

See *FreeFem++* manual for details.

Anyway, for a first approach to *FreeFem++*, we recommend *FreeFem-cs*, `http://www.ann.jussieu.fr/~lehyaric/ffcs/`. *FreeFem-cs* is an integrated environment providing

FreeFem++, and adequate editor and other characteristics. Of course, advanced users may prefer other options.

2.1 *FreeFem-cs*: an integrated environment for *FreeFem++*

FreeFem-cs (*CS* ← Client/Server) is package which contains both *FreeFem++* and an integrated environment for *FreeFem++* providing an intuitive interface. It adds to *FreeFem++* the following goodies:

- Integrated interface, aimed at making users comfortable.
- Color-coded editor.
- Automatic highlighting of *FreeFem++*.
- Compilation errors, linked back to the EDP source code.
- Integrated graphics area for 2d and 3d.
- Online help including documentation in HTML.
- Multi-platform (Windows–GNU/Linux–MacOS).

2.1.1 Installation

For installation, you can get your preferred version from the "*Download*" link (`http://www.ann.jussieu.fr/~lehyaric/ffcs/install.php`) and follow the specific instructions for each platform (which consist in only a few steps). For instance:

- *Windows*: Execute the installation program and follow usual steps. Once installed, click on the *FreeFem++-cs* icon and start using the application.
- *GNU/Linux* (*Ubuntu*, *Debian* and others): Decompress the `.tgz` in your preferred location (for instance, in the desktop). Run the program *FreeFem++-cs* (located in the folder created when decompress).
- *MacOS*: Decompress the `.zip` file in your preferred location (e.g. in the desktop). Run *FreeFem++cs*.

Exercise 1.

Download *FreeFem-cs* from the web (choose the adequate version for your preferred operative system) and install it.

Exercise 2.

Open the FreeFem++ manual, `http://www.freefem.org/ff++/ftp/freefem++doc.pdf`, and search for recommended editors (in Section 1.1). Choose an editor, install and configure it for use with *FreeFem++*.

2.1.2 First steps with *FreeFem-cs*

FreeFem-cs is composed of three different panels:

1. Editor with syntax highlighting (left).
2. Messages returned by the interpreter (bottom).
3. Graphics generated by our numerical simulation (right). endenumerate

Other Characteristics:

- The FreeFem++ script (program) can be run at any time by clicking in the `Run` buttom (top left), or pressing `Ctrl+Shift+R`.
- The script can be stopped at any time by clicking the `Kill` button (top left).
- Dragging a *FreeFem++* script file into *FreeFem-cs* (icon or editor) makes *FreeFem-cs* edit that script.

In the following exercise, we write a very simple FreeFem++ script which (a) plots a simple mesh in the unit square $[0,1] \times [0,1]$, which is defined by two subintervals (of $[0,1]$) in the x axis and also two subintervals in y.

Exercise 3.

Write the following code and run it. Test that a graphic appears in the right panel and a message is written in the bottom panel.

```
mesh Th = square(2,2); // Declare a mesh object and build it
plot(Th);
cout << "Hello world!" << endl;
```

Former code may result quite familiar to C++ programmers.

2.2 A first realistic example

Now we are going to solve for the first time a PDE system by means of the FEM method. More complex problems are left for further sections. Specifically, here we are going to solve the following example (Poisson equation with homogeneous Dirichlet boundary conditions):

$$\begin{cases} \text{Find } u : \bar{\Omega} \to \mathbb{R} \text{ such that} \\ -\Delta u = f \quad \text{in } \Omega, \\ \qquad u = 0 \quad \text{on } \partial\Omega. \end{cases}$$

For that purpose, we proceed as follows:

Step 1. Express the problem in (discrete) variational formulation:

$$\begin{cases} \text{Find } u : \bar{\Omega} \to \mathbb{R} \text{ such that} \\ -\Delta u = f \quad \text{in } \Omega, \\ \qquad u = 0 \quad \text{on } \partial\Omega. \end{cases} \quad \leadsto \quad \begin{cases} \text{Find } u_h \in X_h \text{ such that} \\ \displaystyle\int_\Omega \nabla u_h \cdot \nabla v_h = \int_\Omega f \cdot v_h, \quad \forall v_h \in X_h. \end{cases}$$

Step 2. Translate the variational formulation into *FreeFem++* language. Supposing that the domain, Ω, is given by the unit circle, we can write the following script :

Example 1: First example: Poisson problem homogeneous Dirichlet conditions

```
border gamma(t=0, 2*pi) { x=cos(t); y=sin(t); } // Define boundary
mesh Th = buildmesh(gamma(20)); // Build mesh with 20 intervals on boundary
fespace Xh(Th,P1); // Define P1 finite element space
Xh u,v; // Build two functions in this space (unknown and test function)
func f = exp(x)*sin(y);  // Right hand side
solve Dirichlet(u,v) =   // Define variational formulation
  int2d(Th) ( dx(u)*dx(v) + dy(u)*dy(v) )
  - int2d(Th) ( f*v )
  + on( gamma, u=0 );
plot(u);  // Show results
```

This piece of code contains the fundamentals of FEM with *FreeFem++*.

1. In lines 1 and 2 we define the circular domain. The technique consists of the parametrization of the boundary. Any domain with parametrizable boundary can be easily introduced in *FreeFem++*. For other domains, one has to use a specific tool for mesh construction and

2. In line 3 we define the FE (finite element) space, $\mathcal{P}_1$–Lagrange in this case, and in line 4 we declare two variables in this space. We intend to use the first one, `u`, as the FE unknown (the `trial` function), and the second one, `v` as the `test` function.

3. In line 5 we define a function. Note that standard variables `x` and `y` are predefined and must not be declared.

4. In lines 6–9 we solve the variational problem. Note that those lines constitute a quasi-literal transcription of the variational problem formulated in `Step 1`. Some comments:

   (a) By default, PDE operators like gradient (∇) are not predefined (although they can be defined using macros, as we see in a further section). So one must use the operators `dx` ($\frac{\partial}{\partial x}$) and `dy` ($\frac{\partial}{\partial y}$). For 3D programs, also `dz` can be employed.

   (b) Dirichlet conditions are imposed as the "artificial" sum of a term to the bilinear form.

5. Finally, in line 10 we plot the obtained solution. Scalar data (as, in this case, u) is plotted by contour plots, while vector data is plotted as arrow field (for instance, the velocity unknown in the context of Stokes equations).

2.3 Saving to VTK for high-quality graphics

VTK consists of an open source C++ library for visualization of different types of data (scalar, vector, tensor, etc.). Last versions of FreeFem++ include a module (called `iovtk`) which can be loaded for use VTK. This way users can save any FE function to a `.vtk` file and then employ any of the available advanced applications for manipulation and visualization of the data contained in that file. In section **??** we delve into one of those applications, called Paraview[2].

The following code can be appended to the script above for saving the solution, u, into a VTK file.

```
load "iovtk";
savevtk("/tmp/output.vtk", Th, u, dataname="Temperature");
```

[2] http://www.paraview.org/

The module `iovtk` provides the function `savevtk`. Their compulsory parameters are: (1) name of the output VTK file, (2) name of the mesh, (3) FE function to be saved. More than one function can be saved in the same file, as we will see below. The last parameter is optional (but recommended) and provides a name for each saved function. In this case, assuming that the solution represents the equilibrium state of a heating experiment, the only data set is called "Temperature". We can use this name to access the data in the future (for instance using Paraview).

3 Other Boundary Conditions

In this section we go beyond the Poisson problem and generalize it in different ways.

1. Introducing other kind of boundary conditions (Neumann b.c.)
2. Handling transient (time dependent) problems (Heat equation).

3.1 Poisson Problem With Mixed Neumann/Dirichlet Boundary Conditions

We set the problem: given

- $\Omega \subset \mathbb{R}^2$, with smooth piecewise boundary, where we distinguish two zones: $\partial\Omega = \Gamma_0 \cup \Gamma_1$
- $\nu > 0, \quad f : \Omega \to \mathbb{R}, \quad g : \Gamma_0 \to \mathbb{R}$

Given $f : \Omega \to \mathbb{R}$, $g_0 : \Gamma_0 \to \mathbb{R}$ and $g_2 : \Gamma_1 \to \mathbb{R}$, we try to find $u : \bar{\Omega} \to \mathbb{R}$ such that:

$$\begin{cases} -\nu\Delta u = f & \text{in } \Omega, \\ u = g_0 & \text{on } \Gamma_0, \\ \dfrac{\partial u}{\partial n} = g_1 & \text{on } \Gamma_1. \end{cases} \tag{1}$$

Then one has have a (non homogeneous) Dirichlet b.c. in Γ_0 and a Neumann b.c on Γ_1. Remember that last condition means, means $\nabla u \cdot n = g_1$, where n is the exterior normal vector.

- The theory for **non-homogeneous Dirichlet** conditions, $u|_{\Gamma_0} = g_0$, is based on writing the solution as

  $$u = u_0 + u_D,$$

  where u_0 is a solution of the homogeneous problem ($u|_{\Gamma_0} = 0$) while u_D verifies $u|_{\Gamma_0} = g_0$. In practice, for Dirichlet conditions one proceed as follows:

1. Build the FE linear system $Ax = b$, where A comes from a bilinear form, $a(\cdot,\cdot)$ and b comes from a linear form, $L(\cdot)$. Both of A and b are, typically constructed by quadrature formulae in triangles.
2. Select the rows of A and b which correspond to equations relative to degrees of freedom (for instance vertices of the triangles) placed on Γ_D. Then modify them, imposing explicitly the value of u on that degrees of freedom.

This issue is automatized by *FreeFem++* and then we are not going deeper.

- But Neumann boundary conditions appear in a natural way in the variational formulation. Specifically, when the Green formula (integration by parts) is applied, one gets the following problem:

$$\begin{cases} \text{Find } u_h \in U_h \text{ such that} \\ \displaystyle\int_\Omega \nabla u_h \cdot \nabla v_h = \int_\Omega f\, v_h + \int_{\Gamma_1} g_1\, v_h \quad \text{for each } v_h \in U_h, \end{cases}$$

being U_h the set of functions $u_h : \Omega \to \mathbb{R}$ such that

- $u_h|_T \in \mathbb{P}_k[x]$ (polynomials of degree k) for all $T \in \mathcal{T}_h$ and
- $u_h|_{\Gamma_0} = 0$.

3.1.1 FreeFem++ Example

Here we show a FreeFem++ program for the Poisson problem presented above Note that here we use the keyword `problem` for defining the variational problem, which is solved later (instead of `solve`, which was used in Example 1 for defining and solving the problem).

Example 2: Poisson problem with mixed Dirichlet/Neumann boundary conditions

```
// 1. Pre-proceso

// 1.1. Mesh
border gamma0(t=2*pi, 0) { x=1.5*cos(t); y=sin(t); }
border gamma1(t=0, 2*pi) { x=4*cos(t); y=4*sin(t); }
mesh Th = buildmesh(gamma1(40)+gamma0(30));
plot(Th, wait=1);

// 1.2. FE space and functions
fespace Vh(Th,P1);
```

```
Vh u,v;

// 1.3. Definition of data
real nu=0.3;
func f=8*(x^2+y^2);
func g=400;

// 2.- Defining the problem and solving it

problem PoissonDirNeu(u,v) =
  // Bilinear form:
  int2d(Th)( nu*( dx(u)*dx(v) + dy(u)*dy(v) ))
  // Linear form:
  - int2d(Th)( f*v )
  - int1d(Th, gamma1)( nu*g*v )
  // Dirichlet boundary condition
  + on(gamma0, u=g);

PoissonDirNeu; // Mount linear system and solve the problem

// 3. Post-processing
plot(u, value=1, fill=1, wait=1);
```

3.2 Poisson Problem With Robin Boundary Conditions

Let us consider the differential problem, which generalizes (1):

$$\begin{cases} -\nu\Delta u = f & \text{in } \Omega, \\ u = g_0 & \text{on } \Gamma_0, \\ au + b\dfrac{\partial u}{\partial n} = g_1 & \text{on } \Gamma_1, \end{cases} \tag{2}$$

where $a, b \in \mathbb{R}$. For $a, b \neq 0$, the third equation in (2) is termed a *Robin boundary condition.*

Integration by parts one can obtain the following variational formulation:

$$\begin{cases} \text{Find } u_h \in U_h \text{ such that} \\ \displaystyle\int_\Omega \nabla u_h \cdot \nabla v_h + \frac{a}{b}\int_{\Gamma_1} uv = \int_\Omega f\, v_h + \frac{1}{b}\int_{\Gamma_1} g_1\, v_h \quad \text{for each } v_h \in U_h, \end{cases}$$

being U_h defined as above.

Exercise 4.

Develop a FreeFem++ script for the finite element approximation of the solution of the problem presented above. For instance, use $a = 1, b = 1$ and the same domain and data as in Example 2.

4 Delving into the FreeFem++ language

4.1 Data Types, Arrays and Matrices

Fundamental data types are similar to C++. But some fundamental types of C++ are not present in FreeFem++ (and vice versa) for instance:

Example 3: FreeFem++ fundamental data types and operations

```
// 1) Integers

int i;
i = 10;
int j =-10;
// unsigned k=20; // Error the identifier usigned does not exist
cout << i << ", " << j << endl;

// Some arithmetics
cout << " i+j:" << i+j << endl;
cout << "max(i,j): " << max(i,j) << endl;
cout << "square(i): " << i^2 << " or " << square(i) << endl;
cout << "sqrt(abs(j)): " << sqrt(abs(j)) << endl;

// 2) Floatting point numbers

real x, y; // Double precisson numbers (termed double in C++)
x = 0.001;
y = pi; // Pi is a pre-defined keyword

// 3) Also complex
complex c = 1-2i;
cout << "c=" << c << "... " << 2*c-2 << endl;

// 4) Characters and strings
// char s = 'a' // // Error the identifier char does not exist
```

```
string s = "a";
string t = "Esto es una cadena de caracteres";
cout << s << ", " << t << endl;
cout << "Uni\'on " << ", " << s+t << endl;
cout << "Subconjunto: " << t(0:3) << endl;

// 5) Arrays
cout << endl << endl;

real [int] v(10);  // array of 10 real
v = 1.03; // set all the array to 1.03
v[1]=2.15;
cout << v[0] << " " << v[9] << " size of v = "
  << v.n << " min: " << v.min << " max:" << v.max
  << " sum : " << v.sum << endl;
// change the size of array
v.resize(12);
v(10:11)=3.14;
v(5:9)=sqrt(2);
cout << " resized v: " << v << endl;
real[int] w1(12), w2(12);
w1 = 2*v;
w2 = w1+v;
cout << "w2:" << w1 << endl;
cout << "min: " << v.min << ", sum: " << v.sum << endl;

// Arrays with two indexes

int N=3;
real[int,int] A(N,N); // Squared NxN matrix
real[int] b1(N), b2(N);
b1=[4,5,6];
b2=[1,2,3];
A=1; // Fill A with ones
A(:,1)=2; // Fill first column
cout << A << endl;
cout << b1'*A << endl; // b^T times A
cout << b1'*b2 << endl; // Scalar product

// Sparse matrices

```

```
matrix M = A; // Now M is a sparse matrix
cout << "Storage: for each nozero value, row column value(row,
    ↪ column)" << endl;
cout << M << endl;
```

4.2 Linear System Associated to Variational Formulation

In *FreeFem++*, one can use the keyword `varf` to store the matrix and vector related to a variational formulation. Then the operator `^-1*` can be used to solve the associated linear system.

The advantage of using this procedure is that it is faster that `solve` or `problem` (about 4 times faster, according to *FreeFem++* documentation).

The following script uses `varf` for solving Example 2.

Exercise 5.

Develop a FreeFem++ script for the finite element approximation of the solution of the problem presented above. For instance, use $a = 1, b = 1$ and the same domain and data as in Example 2.

Example 4: Linear System Associated to Variational Formulation

```
// 1. Pre-proceso

// 1.1. Mesh
mesh Th = square(4, 4);
plot(Th, wait=1);
int[int] dirichletBoundary = [1,2,3,4];

// 1.2. FE space and functions
fespace Vh(Th,P1);
Vh u,v;

// 1.3. Definition of data
real nu=1;
func f=4;
func g=0;

// 2.- Defining the problem

```

```
varf PoissonDirichletVarf(u,v) =
  // Bilinear form:
  int2d(Th)( nu*( dx(u)*dx(v) + dy(u)*dy(v) ))
  // Linear form:
  - int2d(Th)( f*v )
  // Dirichlet boundary condition
  + on(dirichletBoundary, u=g);

matrix A = PoissonDirichletVarf (Vh, Vh); // Mount sparse matrix
real[int] b = PoissonDirichletVarf(0, Vh); // Mount RHS

u[] = A^-1 * b;

// 3. Post-processing
plot(u, value=1, fill=1, wait=1);
```

5 Solving Evolution Equations

5.0.1 The Heat Equation

Former examples were steady, namely time independent. Now we set the first transient problem, where the time variable is present. given

- $\Omega \subset \mathbb{R}^2$, with smooth piecewise boundary, $\partial\Omega = \Gamma_0 \cup \Gamma_1$
- $T > 0$: final time, n: number of time iterations in $[0, T]$.
- u_0: $\Omega \to \mathbb{R}$: temperature at initial time.
- $\nu > 0$, $f : \Omega \times (0, T) \to \mathbb{R}$ (heat source in the domain). $u_{\text{ext}} : \Gamma_1 \times (0, T) \to \mathbb{R}$ (heat source on boundary Γ_1).

For time discretization, consider $n + 1$ time instants in $[0, T]$, $t_k = \texttt{dt} \cdot k$, $k = 0, ..., n$, being $\texttt{dt} = T/n$ the time step. The *Implicit Euler* method reads:

- Initialization: for $k = 0$, take $u^0 = u(t = 0) = u_0$
- Step k: given $u(t_k)$, find $u^{k+1} \in U_h$ (defined in section 3.1) such that

$$a(u_h, v_h) = b(v_h) \quad \forall v_h \in U_h,$$

where

$$a(u,v) = \int_\Omega \frac{u^{k+1}}{dt} v + \nu \int_\Omega \nabla u \cdot \nabla v,$$
$$b(v) = \int_\Omega f \cdot v + \int_\Omega \frac{u^k}{dt} v$$

5.0.2 *FreeFem++* program

Example 5: Stokes Equations

```
load "iovtk"; // We will output vtk

// 1. Pre-processing

// 1.1. Mesh
real R=1;
border gamma0(t=0, pi/4)     { x=R*cos(t); y=R*sin(t); }
border gamma1(t=pi/4, 2*pi) { x=R*cos(t); y=R*sin(t); }

int n=30;
mesh Th = buildmesh(gamma0(n)+gamma1(9*n));
plot(Th, wait=1);

// 1.2. FE space and functions
fespace Vh(Th,P1);
Vh u, v;
Vh uold;

macro gradient(u) [dx(u), dy(u)] // End Of Macro

// 1.3. Data definition
real nu=1;
real t=0, T=1;  // Time interval [0,T]
int  N=100; // Number of time iterations
real dt=T/N;  // Time step

func f=0; //8*(x^2+y^2);
// func real g1(real x, real y, real t) {
//   return 40*t;
// }
```

```
func real g0(real x, real y, real t) {
  return 100*(1-1./(t+1));
}
func real g1(real x, real y, real t) {
  return 0;
}

func u0=0; // Init

uold = u0;

// 2. Processing

// Declare (but not solve) the heat equation variational problem
problem heatEquation(u,v)=
  // Bilineal form:
  int2d(Th)(
    u*v/dt +
    nu*gradient(u)'*gradient(v) // ' means transpose
  )
  // Linear form
  - int2d(Th)( uold*v/dt + f*v )
  - int1d(Th, gamma1) ( g1(x,y,t)*v ) // Neumann boundary condtion

  // Dirichlet boundary condtion
  + on(gamma0, u=g0(x,y,t));

// Time iteration loop
for (int k=0; k<N; ++k ) {

  t = t + dt;      // Increase current time
  heatEquation;    // Solve the PDE variational problem
  uold = u;        // Save solution for next time step

  // 3. Post-processing (save to VTK for further displaying with Paraview)
  string filename="/tmp/heat_equation-" + k + ".vtk";
  savevtk(filename, Th, u, dataname="Temperature");
}
```

5.1 The Stokes equations

The Stokes equations can be considered as the linear steady version of Navier-Stokes equations (which describe the behaviour of a newtoninan fluid as atmosphere, ocean, flux around vehicles, etc.

$$\begin{cases} -\nu\Delta\mathbf{u} + \nabla p = f \\ \nabla\cdot\mathbf{u} = 0 \\ \qquad + \text{boundary conditions,} \end{cases}$$

where the unknowns are: $\mathbf{u} = (u, v) : \Omega \to \mathbb{R}$ (velocity field of fluid) and $p : \Omega \to \mathbb{R}$ pressure in each point of the domain. Thus the first equation must be understood in vectorial way, specifically, in the 2D case:

$$\Delta u + \partial_x p = f_1,$$
$$\Delta v + \partial_y p = f_2,$$

where $f = (f_1, f_2)$.

In this section we show a usual test for the Stokes 2D simultion, which is know as **cavity test**. This test is usually run in a rectangular domain but, in this case, with the purpose of illustrate the construction in *FreeFem++* of complex parametric geometries, we have introduced some holes in the rectangular domain. They are defined by parametric figures which are known as conchoids[3].

Homogeneous Dirichlet b.c., $(u, v) = (0, 0)$, are imposed for $\mathbf{u}$ on the whole boundary excepting the top line, where we fix $(u, v) = (1, 0)$ (positive horizontal velocity). We use the stable FE combination $\mathcal{P}_2/\mathcal{P}_1$ (polynomials with degree 2 for velocity and degree 1 for pressure).

5.1.1 Programación con ***FreeFem++***

```
// 2D Stokes equations
// Cavity test in a domain with some parametric holes

//,-------------------------------------------------------
//| STEP 1. Defining the domain and meshing it
//`-------------------------------------------------------

// Macro for the 2D boundary defining a hole. They are parametric
// curves called "conchoids". In the macro:
// n = number of 'petals', P = center of the hole
```

[3] http://en.wikipedia.org/wiki/Conchoid_%28mathematics%29

```
int NMAX=20;
macro conchoid(name, n, P, thelabel)
 name(i=0,NMAX) {
    real a=1.0, b=2.0;
    real theta = i*2*pi/NMAX;
    real rho = a * cos(n*theta)+b;
    x = P[0] + rho*cos(theta);
    y = P[1] + rho*sin(theta);
    label = thelabel;
} // EOM

// Definition of some conchoids
border conchoid(c2,2,[0,0]   ,0);
border conchoid(c3,3,[-10,0] ,0);
border conchoid(c4,4,[0,0]   ,0);
border conchoid(c5,5,[10,0]  ,0);
border conchoid(c6,6,[0,0]   ,0);
border conchoid(c7,7,[0,0]   ,0);

// External rectangle
real xcoor = 15, ycoor = 5;
border lx1(k=-xcoor,xcoor) { x=k; y=-ycoor; label=1; }
border lx2(k=-xcoor,xcoor) { x=k; y=+ycoor; label=3; }
border ly1(k=-ycoor,ycoor) { x=-xcoor; y=k; label=2; }
border ly2(k=-ycoor,ycoor) { x=+xcoor; y=k; label=2; }

int nx=40, ny=20, nc=50;
mesh Th = buildmesh( ly1(-ny)+lx1(nx)+ly2(ny)+lx2(-nx)
    + c3(-nc) + c4(-nc) + c5(-nc) );

//,----------------------------------------------------------
//| STEP 2. Resolution of Stokes problem in previous domain
//`----------------------------------------------------------

fespace Uh(Th,P2); Uh u,v,uu,vv;  // Velocity functions
fespace Ph(Th,P1); Ph p,pp;       // Pressure functions

real upperVelocity=1;

macro grad(u) [dx(u), dy(u)] // end of macro

```

```
// Definition of Stokes problem

problem stokes2d( [u,v,p], [uu,vv,pp], solver=LU) =
    int2d(Th)(
        grad(u)'*grad(uu) + grad(v)'*grad(vv)
        + grad(p)'*[uu,vv] + pp*(dx(u)+dy(v)) //'
        - 1e-10*p*pp )
  + on(0,1,2,u=0,v=0) + on(3,u=upperVelocity,v=0);

stokes2d;  // Resolution of Stokes problem

// Save to VTK (for high quality plotting)
load "iovtk";
savevtk("/tmp/stokes.vtk", Th, [u,v,0], p);
```

A Paraview

ParaView is an open source multiple-platform application for interactive, scientific visualization. It was developed to analyze extremely large datasets using distributed memory computing resources. It can be run on supercomputers to analyze datasets of terascale as well as on laptops for smaller data.

For visualization of data, that lives in a mesh where the simulation was performed, there are basically three steps:

1. *Reading* data into Paraview (from a VTK file)
2. *Filtering*, that is applying one or more filters in order to generate, extract or derive features from data.
3. *Rendering* an image from the data and adjusting the viewing parameters for improve the final visualization.

This tree steps are controlled through a panel in the right, called `Pipeline browser`. The pipeline concept consists on a chain of modules, starting from the data stored in a file. Each of them takes in some data, operates on it and presents the result in a dataset. From the Paraview users guide:

> "Reading data into ParaView is often as simple as selecting `Open` from the *File* menu, and then clicking the glowing *Accept* button on the reader's *Object Inspector* tab. ParaView comes with support for a large number of file formats, and its modular architecture makes it possible to add new file readers. Once a file is read, ParaView automatically renders it in

a view. In ParaView, a view is simply a window that shows data. There are different types of views, ranging from qualitative computer graphics rendering of the data to quantitative spreadsheet presentations of the data values as text. ParaView picks a suitable view type for your data automatically, but you are free to change the view type, modify the rendering parameters of the data in the view, and even create new views simultaneously as you see fit to better understand what you have read in. Additionally, high-level meta information about the data including names, types and ranges of arrays, temporal ranges, memory size and geometric extent can be found in the *Information* tab."

Advanced data processing can be done using the Python Programmable filter with VTK, NumPy, SciPy and other Python modules.

For further details:

1. Video showing how to use FreeFem++ and Paraview for visualization of 2D and 3D cavity tests for the Stokes Equations (partially in spanish). `https://www.youtube.com/watch?v=wChDeo2A03E`

2. Paraview Wikipedia page (in which this appendix is based). `http://en.wikipedia.org/wiki/ParaView`.

3. Resources in the web, for instance `http://vis.lbl.gov/NERSC/Software/paraview/docs/ParaView.pdf`.

4. The paraview users guide (how to unleash the beast!) `http://denali.princeton.edu/Paraview/ParaViewUsersGuide.v3.14.pdf`